Exam 1

Question 1

Data Center Reliability

```
Show[Graphics[{
    Circle[{-18, 0}, 1],
    Circle[{-14, 0}, 1],
    Circle[{-10, 0}, 1],
    Circle[{-6, 0}, 1],
    Circle[{-2, 0}, 1],
    Circle[{2, 2}, 1],
    Circle[{2, -2}, 1],

    Text["1", {-18, 0}],
    Text["2", {-14, 0}],
    Text["3", {-10, 0}],
    Text["4", {-6, 0}],
    Text["5", {-2, 0}],
    Text["6", {2, 2}],
    Text["7", {2, -2}],

    Line[{{-20, 0}, {-19, 0}}],
    Line[{{-17, 0}, {-15, 0}}],
    Line[{{-13, 0}, {-11, 0}}],
    Line[{{-9, 0}, {-7, 0}}],
    Line[{{-5, 0}, {-3, 0}}],
    Line[{{-1, 0}, {0, 0}}],
    Line[{{4, 0}, {5, 0}}],

    Line[{{0, -2}, {0, 2}}],
    Line[{{4, -2}, {4, 2}}],

    Line[{{0, 2}, {1, 2}}],
    Line[{{3, 2}, {4, 2}}],
    Line[{{0, -2}, {1, -2}}],
    Line[{{3, -2}, {4, -2}}]
  }]]
```




```
ClearAll["Global`*"]
dist = BernoulliDistribution[p];
ℛdatacenter =
  ReliabilityDistribution[X1 ⋀ X2 ⋀ X3 ⋀ X4 ⋀ X5 ⋀ (X6 ∨ X7), {{X1, dist},
    {X2, dist}, {X3, dist}, {X4, dist}, {X5, dist}, {X6, dist}, {X7, dist}}];
r[p_] = FullSimplify[Mean[ℛdatacenter]]
```

$-(-2+p)\,p^6$

```
Plot[{p^6 (2 - p^6), r[p]}, {p, 0, 1},
 PlotLegends → Placed[{"UpperLim", "Reliability-iid"}, Right],
 PlotStyle → {Blue, Green}]
```




```
ClearAll["Global`*"]
dist = ExponentialDistribution[λ];
dist4 = ErlangDistribution[2, λ];
ℛdatacenter =
  ReliabilityDistribution[X1 ∧ X2 ∧ X3 ∧ X4 ∧ X5 ∧ (X6 ∨ X7), {{X1, dist},
    {X2, dist}, {X3, dist}, {X4, dist4}, {X5, dist}, {X6, dist}, {X7, dist}}];
FullSimplify[SurvivalFunction[ℛdatacenter, t]]
FullSimplify[HazardFunction[ℛdatacenter, t]]
h[t_] = FullSimplify[HazardFunction[ℛdatacenter, t]];
FullSimplify[∂t h[t]]
Mean[ℛdatacenter]
```

$$\begin{cases} 1 & t \le 0 \\ e^{-6 t \lambda} \left(-1 + 2 e^{t \lambda}\right) \text{GammaRegularized}[2, t \lambda] & \text{True} \end{cases}$$

$$\begin{cases} \lambda \left(6 + \frac{1}{1-2 e^{t \lambda}} - \frac{1}{1+t \lambda}\right) & t \ge 0 \\ 0 & \text{True} \end{cases}$$

$$\begin{cases} \text{Indeterminate} & t == 0 \\ \lambda^2 \left(\frac{1}{(1-2 e^{t \lambda})^2} + \frac{1}{-1+2 e^{t \lambda}} + \frac{1}{(1+t \lambda)^2}\right) & t > 0 \\ 0 & \text{True} \end{cases}$$

$$\frac{199}{882 \lambda}$$

```
ClearAll["Global`*"]
λ = 1.;
R[t_] = Exp[-6 λ t] * (1 + λ t) * (2 - Exp[-λ t]);
f[t_] = -∂t R[t];
h[t_] = f[t] / R[t];
Plot[h[t], {t, 0.001, 10.}, AxesOrigin → {0, 0}]
```




```
ClearAll["Global`*"]
λ = 1.;
dist = ExponentialDistribution[λ];
dist4 = ErlangDistribution[2, λ];
ℛdatacenter =
  ReliabilityDistribution[X1 ∧ X2 ∧ X3 ∧ X4 ∧ X5 ∧ (X6 ∨ X7), {{X1, dist},
    {X2, dist}, {X3, dist}, {X4, dist4}, {X5, dist}, {X6, dist}, {X7, dist}}];

p = 0.5;
ξ = Quantile[ℛdatacenter, p]
If[0.5 ≤ 1 - Exp[-1.], Print[{-(p * ξ)/Log[1 - p], -ξ/Log[1 - p]}], Print[{-(p * ξ)/Log[1 - p], ξ}]]
Mean[ℛdatacenter]
```

$\frac{2}{11 \lambda}$

```
0.164423

{0.118606, 0.237213}

0.225624

0.181818
```

Question 2 — Type II/item censoring without replacement

```
ClearAll["Global`*"]
data = {10.2, 89.6, 54.0, 96.0, 23.3, 30.4, 41.2, 0.8, 73.2, 3.6, 28.0, 31.6};
data = Sort[data]

n = 20.;
r = 12;
tau = Sum[data[[i]], {i, 1, r}] + (n - r) * data[[r]]
- (tau / r) * Log[1 - 0.75]

2 * tau
Quantile[ChiSquareDistribution[2 * r], 0.025]
Quantile[ChiSquareDistribution[2 * r], 0.975]

λL = Quantile[ChiSquareDistribution[2 * r], 0.025] / (2 * tau)
λU = Quantile[ChiSquareDistribution[2 * r], 0.975] / (2 * tau)
-1 / λU * Log[1 - 0.75]
-1 / λL * Log[1 - 0.75]

λL = 12.40 / 2499.8
λU = 39.36 / 2499.8
-1 / λU * Log[1 - 0.75]
-1 / λL * Log[1 - 0.75]
```

```
{0.8, 3.6, 10.2, 23.3, 28., 30.4, 31.6, 41.2, 54., 73.2, 89.6, 96.}

1249.9

144.394

2499.8

12.4012

39.3641

0.00496086

0.0157469

88.0361

279.447

0.0049604

0.0157453

88.0452

279.472
```

Question 3

In[13]:=
```
ClearAll["Global`*"]
data = {2, 3, 8, 1, 1, 4, 1, 4, 5, 1};
λ0 = Mean[data]
LCL = Ceiling[Max[0, λ0 - 3 × √λ0]]
UCL = Floor[λ0 + 3 × √λ0]
```

Out[15]= 3

Out[16]= 0

Out[17]= 8

In[30]:=
```
L = 0;
U = 10;
gammaL = 0.038733;
gammaU = 0.591561;
ξ[θ_] = 1 -
    (CDF[PoissonDistribution[λ0 + θ], U] - CDF[PoissonDistribution[λ0 + θ], L - 1]) +
    gammaL * PDF[PoissonDistribution[λ0 + θ], L] +
    gammaU * PDF[PoissonDistribution[λ0 + θ], U];

(*ξstar[0]
  1-(0.9997-0)+0.038733* (0.0498-0)+0.591561* (0.9997-0.9989)*)

ξ[-1]
1 - (1.0000 - 0) + 0.038733 * (0.1353 - 0) + 0.591561 * (1.0000 - 0.9998)

ARL[θ_] = 1 / ξ[θ];
ARL[0]
ARL[-1]
1 / 0.00535889

Plot[{ARL[θ], ARL[θ]}, {θ, -3, 3}]
```

Out[35]= 0.00527284

Out[36]= 0.00535889

Out[38]= 370.371

Out[39]= 189.651

Out[40]= 186.606

Out[41]=



Question 4 — (X-bar,S^2) joint scheme for μ and σ^2

```
n = 4;
μ0 = 0;
σ0 = 1;
incontrolARLindiv = 500.;

distmu = NormalDistribution[0, 1];
γmu = Quantile[distmu, 1 - 1/(2 * incontrolARLindiv)]
ξmu[δ_, θ_] = 1 - (CDF[distmu, (γmu - δ)/θ] - CDF[distmu, (-γmu - δ)/θ]);
ARLmu[δ_, θ_] = 1 / ξmu[δ, θ];
ARLmu[0, 1]

distsigma = ChiSquareDistribution[n - 1];
γsigma = Quantile[distsigma, 1 - 1/incontrolARLindiv]
ξsigma[θ_] = 1 - CDF[distsigma, γsigma/θ^2];
ARLsigma[θ_] = 1/ξsigma[θ];
ARLsigma[1]

ξmusigma[δ_, θ_] = ξmu[δ, θ] + ξsigma[θ] - ξmu[δ, θ] × ξsigma[θ];
ARLmusigma[δ_, θ_] = 1/ξmusigma[δ, θ];
ARLmusigma[0, 1]
ARLmusigma[δ_, θ_] = (ARLmu[δ, θ] * ARLsigma[θ]) / (ARLmu[δ, θ] + ARLsigma[θ] - 1);
shiftmu = 0.1;
ARLmusigma[shiftmu, 1]

ARLmu[shiftmu, 1]
ARLsigma[1]

((1 - ξmu[shiftmu, 1]) × ξsigma[1]) / ξmusigma[shiftmu, 1]
((1 - 1/ARLmu[shiftmu, 1]) × 1/ARLsigma[1]) / (1/ARLmusigma[shiftmu, 1])
((1 - 1/475.1454) * 1/500) / (1/475.1454 + 1/500 - 1/475.1454 × 1/500)
```

```
3.09023

500.

14.7955

500.

250.25

243.878

475.145

500.
```

```
0.48673

0.48673

0.48673
```

Question 5 — Single sampling plan for attributes with RECTIFYING INSPECTION

```
Clear[Evaluate[Context[] <> "*"]];
p1 = 0.005; (* AQL *)
α = 1 - 0.95; (* producer's risk *)
p2 = 0.175; (* LTPD *)
β = 0.15; (* consumer's risk *)

Q[c_, x_] = Quantile[ChiSquareDistribution[2 × (c + 1)], x];
r[c_] = N[Q[c, 1 - β], 5] / N[Q[c, α], 5];
i = 0;
While[r[i] > p2/p1, Print["Do not use acceptance number c=", i, " because r(c)=",
   N[Q[i, 1 - β], 5], "/", N[Q[i, α], 5] , "=", r[i], ">p2/p1=", p2/p1];
  i++]
Print["Use the acceptance number c=", i, " because r(c)=",
  N[Q[i, 1 - β], 5], "/", N[Q[i, α], 5] , "=", r[i], "≤p2/p1=", p2/p1]
samplesize[c_] = Ceiling[Q[i, 1 - β] / (2 × p2)];
If[Ceiling[Q[i, 1 - β] / (2 × p2)] ≤ Floor[Q[i, α] / (2 × p1)],
  Print["Use the sample size n=", samplesize[i], "."],
  Print["Houston, we have a problem!"]]
```

```
Do not use acceptance number c=0 because r(c)=3.79424/0.102587=36.9857>p2/p1=35.

Use the acceptance number c=1 because r(c)=6.74488/0.710723=9.49017≤p2/p1=35.

Use the sample size n=20.
```

```
(* Unnecessary verification *)
ntot = 1000; (* lot size *)
exactdist[p_] = HypergeometricDistribution[samplesize[i], Round[ntot × p], ntot];
If[CDF[exactdist[p1], i] ≥ 1 - α && CDF[exactdist[p2], i] ≤ β,
  Print["The single sampling plan for attributes
     complies with the producer's and consumer's risk points."],
  Print["The single sampling plan for attributes does not comply
     with the producer's or the consumer's risk points."]]
```

```
The single sampling plan for attributes
  complies with the producer's and consumer's risk points.
```

```
(* Average Outgoing Quality (AOQ) or percentage of defective due to rectifying inspection in a
 single sampling plan and using the binomial approximation to the acceptance probability *)
ATI[n_, c_, p_] = n * CDF[BinomialDistribution[n, p], c] +
    ntot * (1 - CDF[BinomialDistribution[n, p], c]);
midp = 0.1;
ATI[samplesize[i], i, midp]
(* Results using the tables *)
20 * 0.3917 + 1000 * (1 - 0.3917)
```

616.088

616.134

Question 5 — Single sampling plan for variables with UNKNOWN STANDARD DEVIATION

Exercício 7(b)

```
Clear[Evaluate[Context[] <> "*"]];
p1 = 0.01; (* AQL *)
α = 1 - 0.95; (* producer's risk *)
p2 = 0.05; (* LTPD *)
β = 0.1; (* consumer's risk *)

ns = 59;
ks = 1.937713;

gdist = NormalDistribution[0, 1];
Ω[x_] := Quantile[gdist, x];
```

$$\text{Round}\left[\frac{\Omega[1-p_1] - k_s \times \sqrt{\frac{3\times n_s - 4}{3\times n_s - 3}}}{\sqrt{\frac{1+\frac{3\times n_s \times k_s^2}{6\times n_s - 8}}{n_s}}}, 0.01\right]$$

```
CDF[NormalDistribution[0, 1], %]
```

$$\text{Round}\left[\frac{\Omega[1-p_2] - k_s \times \sqrt{\frac{3\times n_s - 4}{3\times n_s - 3}}}{\sqrt{\frac{1+\frac{3\times n_s \times k_s^2}{6\times n_s - 8}}{n_s}}}, 0.01\right]$$

```
CDF[NormalDistribution[0, 1], %]
```

1.77

0.961636

-1.29

0.0985253

```
U = 3;
sigma = 0.1;
dist = NormalDistribution[U + 2 * sigma, sigma];
data = RandomVariate[dist, ns + 1];
xbar = Round[Mean[data], 0.001]
sx = Round[√Variance[data] , 0.001]
(U - xbar)/sx
If[(U - xbar)/sx ≥ ks, Print["We should accept the lot."],
 Print["We should reject the lot."]]
```

3.207

0.12

-1.725

We should reject the lot.